
begin model      #All concentrations are in nM units 
 begin parameters 
  Temp   37  #Temperature of reaction. Affects rate of intein splicing 
  ppt     1000  #Total parts per trillion of estrogen 
  rfp_copies                         15  #Number of copies of rfp containing plasmid per cell 
  int_copies        2984  #Number of copies of intein containing plasmids per cell 
  estrogen_out ppt * 5.6e-02  #Total concentration of estrogen outside cell  
  estrogen_in  0  #Total concentration of estrogen molecules inside cell  
  t7_intein  0  #Total concentration of t7-intein complexes 
  t7_estrogen_u  0  #Total concentration of unspliced T7-estrogen complexes 
  t7_estrogen_s  0  #Total concentration of spliced T7-estrogen complexes 
  t7_nc   0  #Total concentration of completed t7 polymerase 
  plas_int  int_copies * 1.0e-03 #Total concentration of T7-intein plasmids 
  plas_rfp rfp_copies * 1.0e-03 #Total concentration of rfp plasmid 
  t7_rfp   0  #Total concentration of t7 bound to plasmid 
  mrna_rfp  0  #Total concentration of mRNAs to code for RFP 
  prot_rfp  0  #Total concentration of RFP proteins  
  mrna_int  0  #Total concentration of mRNAs to code for T7-Intein 
  prot_yfp  0  #Total concentration of YFP proteins 
  
  k_e    1.7e-02  #Rate at which estrogen diffuses through membrane 
  k_e_t7_on  1.3e-03  #Rate at which estrogen binds to T7 intein 
  k_e_t7_off  1.2e-03  #Rate at which estrogen dissociates from T7 intein 
  k_splice   7.1e-04  #Rate at intein splices out 
  k_t7_rfp_on  3.3e-01  #Rate at which T7 binds to RFP plasmid 
  k_t7_rfp_off  1.0e-01  #Rate at which T7 dissociates from RFP plasmid 
  k_rna_rfp  8.8e-01  #Rate at which rna is synthesized from RFP plasmid 
  k_rna_deg_rfp  4.3e-03  #Rate at which rna of RFP degrades 
  k_prot_rfp  9.0e03  #Rate at which RFP protein is made 
  k_prot_deg_rfp  8.3e-04  #Rate at which RFP proteins degrades 
  k_rna_int  2.6e-02  #Rate at which rna is synthesized from T7-Intein plasmid 
  k_rna_deg_int  4.3e-03  #Rate at which rna of T7-intein degrades 
  k_prot_int  2.2e-03  #Rate at which T7-intein is made from mRNA 
  k_prot_deg_int   9.7e-04  #Rate at which T7 degrades 
 end parameters 
 
 begin molecule types 
  E(l~U~B,L~I~O)    #Estrogen outside of cell, can be inside or outside cell 
  T7_intein(n,r,Y~A~S,c)   #T7 intein. Can be attached or spliced 
  T7_n(n,b)    #N-terminal t7 
  T7_c(c)     #C-terminal t7 
  RFP_plasmid(p,A~0~1)   #Plasmid that codes for RFP 
  mRNA_RFP()    #mRNA that codes for RFP 
  protein_RFP()    #RFP protein 
  INT_plasmid()    #Plasmid that codes for T7-Intein and YFP 
  mRNA_INT()    #mRNA that codes for T7-Intein and YFP 
  protein_YFP()    #YFP protein  
 end molecule types 
 
 begin seed species 
  E(l~U,L~O)       estrogen_out  
  E(l~U,L~I)       estrogen_in   
  T7_n(n!1,b).T7_intein(n!1,r,Y~A,c!2).T7_c(c!2)      t7_intein  



  T7_n(n!1,b).E(l!3~B,L~I).T7_intein(n!1,r!3,Y~A,c!2).T7_c(c!2)  t7_estrogen_u      
  E(l!3~B,L~I).T7_intein(n,r!3,Y~S,c)    t7_estrogen_s    
  T7_n(n!4,b).T7_c(c!4)      t7_nc 
  RFP_plasmid(p,A~0)      plas_rfp 
  RFP_plasmid(p!5,A~1).T7_n(n!4,b!5).T7_c(c!4)   t7_rfp 
  mRNA_RFP()       mrna_rfp 
  protein_RFP()       prot_rfp 
  INT_plasmid()       plas_int 
  mRNA_INT()       mrna_int 
  protein_YFP()       prot_yfp 
 end seed species 
 
 begin observables 
  Molecules E_out    E(l~U,L~O)       
  Molecules E_in    E(l~U,L~I)       
  Molecules   T7_unbound   T7_n(n!1,b).T7_intein(n!1,r,Y~A,c!2).T7_c(c!2)    
  Molecules    T7_estrogen_u  T7_n(n!1,b).E(l!3~B,L~I).T7_intein(n!1,r!3,Y~A,c!2).T7_c(c!2)  
  Molecules T7_estrogen_s  E(l!3~B,L~I).T7_intein(n,r!3,Y~S,c)    
   
  Molecules  T7_nc    T7_n(n!4,b).T7_c(c!4) 
  Molecules    RFP_Plas_U   RFP_plasmid(p,A~0) 
  Molecules  T7_RFP    RFP_plasmid(p!5,A~1).T7_n(n!4,b!5).T7_c(c!4)  
  Molecules  mRNA_RFP  mRNA_RFP() 
  Molecules  prot_RFP  protein_RFP() 
  Molecules INT_Plas  INT_plasmid() 
  Molecules mRNA_INT  mRNA_INT() 
  Molecules  prot_YFP  protein_YFP() 
 end observables 
 
 begin functions 
  mRNA_RFP_synth() = k_rna_rfp * T7_RFP 
  mRNA_RFP_deg() = k_rna_deg_rfp 
  protein_RFP_synth() = k_prot_rfp * mRNA_RFP * sqrt(1/(1 + prot_RFP)) 
  mRNA_INT_synth() = k_rna_int * INT_Plas 
  protein_INT_synth() = k_prot_int * mRNA_INT * sqrt(1/(1 + T7_unbound)) 
  #Rate of YFP synth = Rate of intein synthesis  
  protein_YFP_synth() = k_prot_int * mRNA_INT * sqrt(1/(1 + prot_YFP))  
  int_Splice() = k_splice * 1/(1 + sqrt(abs(37 - Temp))) 
  e_in() = k_e * (E_out)/(E_out + E_in) 
  e_out() = k_e * (E_in)/(E_out + E_in) 
 end functions 
 
 begin reaction rules 
  #Describes Rate at which mRNA T7-Intein is transcribed/degraded 
  0 <-> mRNA_INT()      mRNA_INT_synth(), k_rna_deg_int 
 
  #Describes Rate at which T7-Intein is translated/degraded 
  0 <-> T7_n(n!1,b).T7_intein(n!1,r,Y~A,c!2).T7_c(c!2)  protein_INT_synth(), k_prot_deg_int 
 
  #Describes Rate at which T7 polymerase is degraded 
  T7_n(n!4,b).T7_c(c!4) -> 0      k_prot_deg_int 
 
  #Describes estrogen entering and leaving cell 



  E(l~U,L~O) <-> E(l~U,L~I)       e_in(), e_out() 
 
  #Describes rate of T7-estrogen association/disassociation    
  E(l~U,L~I) + T7_n(n!1,b).T7_intein(n!1,r,Y~A,c!2).T7_c(c!2) <->         
  T7_n(n!1,b).E(l!3~B,L~I).T7_intein(n!1,r!3,Y~A,c!2).T7_c(c!2)       k_e_t7_on, k_e_t7_off 
 
  #Describes the rate at which the intein is spliced out and t7 is formed 
  T7_n(n!1,b).E(l!3~B,L~I).T7_intein(n!1,r!3,Y~A,c!2).T7_c(c!2) ->  
  E(l!3~B,L~I).T7_intein(n,r!3,Y~S,c) + T7_n(n!4,b).T7_c(c!4)   int_Splice()  
             
  #Describes the rate at which T7 binds to promoter of plasmid 2 
  T7_n(n!4,b).T7_c(c!4) + RFP_plasmid(p,A~0) <->  
  RFP_plasmid(p!5,A~1).T7_n(n!4,b!5).T7_c(c!4)     k_t7_rfp_on, k_t7_rfp_off 
 
  #Describes the rate at which the mRNA of RFP is synthesized/degraded 
  0 <-> mRNA_RFP()      mRNA_RFP_synth(), mRNA_RFP_deg() 
 
  #Describes the rate at which RFP is synthesized/degraded 
  0 <-> protein_RFP()      protein_RFP_synth(), k_prot_deg_rfp  
 
  #Describes the rate at which YFP is synthesized 
  0 <-> protein_YFP()      protein_YFP_synth(), k_prot_deg_rfp 
 end reaction rules  
end model 
 
## actions ## 
#construct reaction network 
generate_network({overwrite=>1}) 
 
#kinetics 
saveConcentrations(); 
simulate({method=>"ode",suffix=>"ssa",t_end=>7200,n_steps=>7200,atol=>1e-12,rtol=>1e-12,sparse=>1}) 
resetConcentrations(); 


